Problem:
Score : 400 points

Problem Statement

There is a grid of square cells with H horizontal rows and W vertical columns. The cell at the i-th row and the j-th column will be denoted as Cell (i, j).

In Cell (i, j), a_{ij} coins are placed.

You can perform the following operation any number of times:

Operation: Choose a cell that was not chosen before and contains one or more coins, then move one of those coins to a vertically or horizontally adjacent cell.

Maximize the number of cells containing an even number of coins.

Constraints

All values in input are integers.

1 \leq H, W \leq 500

0 \leq a_{ij} \leq 9

Input

Input is given from Standard Input in the following format:

H W
a_{11} a_{12} ... a_{1W}
a_{21} a_{22} ... a_{2W}
:
a_{H1} a_{H2} ... a_{HW}

Output

Print a sequence of operations that maximizes the number of cells containing an even number of coins, in the following format:

N
y_1 x_1 y_1' x_1'
y_2 x_2 y_2' x_2'
:
y_N x_N y_N' x_N'

That is, in the first line, print an integer N between 0 and H \times W (inclusive), representing the number of operations.

In the (i+1)-th line (1 \leq i \leq N), print four integers y_i, x_i, y_i' and x_i' (1 \leq y_i, y_i' \leq H and 1 \leq x_i, x_i' \leq W), representing the i-th operation. These four integers represents the operation of moving one of the coins placed in Cell (y_i, x_i) to a vertically or horizontally adjacent cell, (y_i', x_i').

Note that if the specified operation violates the specification in the problem statement or the output format is invalid, it will result in Wrong Answer.

Sample Input 1

2 3
1 2 3
0 1 1

Sample Output 1

3
2 2 2 3
1 1 1 2
1 3 1 2

Every cell contains an even number of coins after the following sequence of operations:

Move the coin in Cell (2, 2) to Cell (2, 3).

Move the coin in Cell (1, 1) to Cell (1, 2).

Move one of the coins in Cell (1, 3) to Cell (1, 2).

Sample Input 2

3 2
1 0
2 1
1 0

Sample Output 2

3
1 1 1 2
1 2 2 2
3 1 3 2

Sample Input 3

1 5
9 9 9 9 9

Sample Output 3

2
1 1 1 2
1 3 1 4

Solution in Common Lisp (SBCL 1.1.14) (judge verdict: Accepted)

(eval-when (:compile-toplevel :load-toplevel :execute)
  (defparameter OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0)))
  #+swank (progn (ql:quickload '(:cl-debug-print :fiveam))
                 (shadow :run)
                 (use-package :fiveam)))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)

;; BEGIN_INSERTED_CONTENTS
(declaim (inline split-ints-into-vector))
(defun split-ints-into-vector (string dest-vector &key (offset 0))
  (declare (string string)
           ((simple-array * (*)) dest-vector)
           ((integer 0 #.most-positive-fixnum) offset))
  (loop for idx from offset below (length dest-vector)
        for pos1 = 0 then (1+ pos2)
        for pos2 = (position #\space string :start pos1 :test #'char=)
        do (setf (aref dest-vector idx)
                 (parse-integer string :start pos1 :end pos2))
        finally (return dest-vector)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(defmacro println (obj &optional (stream '*standard-output*))
  `(let ((*read-default-float-format* 'double-float))
     (prog1 (princ ,obj ,stream) (terpri ,stream))))

;; Hauptteil

(defmacro with-collecting (&body body)
  (let ((list (gensym))
        (head (gensym))
        (length (gensym)))
    `(let* ((,list (list :dummy-element))
            (,head ,list)
            (,length 0))
       (labels ((collect (x)
                  (setf (cdr ,list) (list x))
                  (setf ,list (cdr ,list))
                  (incf ,length)))
         ,@body
         (values (cdr ,head) ,length)))))

(defun main ()
  (let* ((h (read))
         (w (read))
         (matrix (make-array (list h w) :element-type 'uint4))
         (row (make-array w :element-type 'uint4)))
    (dotimes (i h)
      (split-ints-into-vector (read-line) row)
      (dotimes (j w)
        (setf (aref matrix i j) (aref row j))))
    (multiple-value-bind (operations n)
        (with-collecting
          (dotimes (i h)
            (cond ((zerop i)
                   (loop for j from 0 below (- w 1)
                         when (oddp (aref matrix i j))
                         do (decf (aref matrix i j))
                            (incf (aref matrix i (+ j 1)))
                            (collect (list i j i (+ j 1)))))
                  ((oddp i)
                   (when (oddp (aref matrix (- i 1) (- w 1)))
                     (decf (aref matrix (- i 1) (- w 1)))
                     (incf (aref matrix i (- w 1)))
                     (collect (list (- i 1) (- w 1) i (- w 1))))
                   (loop for j from (- w 1) above 0
                         when (oddp (aref matrix i j))
                         do (decf (aref matrix i j))
                            (incf (aref matrix i (- j 1)))
                            (collect (list i j i (- j 1)))))
                  ((evenp i)
                   (when (oddp (aref matrix (- i 1) 0))
                     (decf (aref matrix (- i 1) 0))
                     (incf (aref matrix i 0))
                     (collect (list (- i 1) 0 i 0)))
                   (loop for j from 0 below (- w 1)
                         when (oddp (aref matrix i j))
                         do (decf (aref matrix i j))
                            (incf (aref matrix i (+ j 1)))
                            (collect (list i j i (+ j 1))))))))
      (println n)
      (dolist (op operations)
        (destructuring-bind (i1 j1 i2 j2) op
          (format t "~A ~A ~A ~A~%" (+ i1 1) (+ j1 1) (+ i2 1) (+ j2 1)))))))

#-swank(main)